7.1

7.1 $e_{nw} = 6\,nV/\sqrt{Hz}$; $20^2 = 6^2(f_{ce}/10+1) \Rightarrow$ $f_{ce} \cong 100$ Hz. $E_n = 6\times10^{-9}\left[100\ln(10^6/10^3)+10^6\right]^{1/2}$ $= 6\,\mu V$ (rms).

7.2 $A_{total} = \dfrac{A_0^2}{(1+jf/f_B)^2}$; $|A_{total}| = \dfrac{A_0^2}{1+(f/f_B)^2}$.

$$NEB = \frac{1}{A_0^4}\int_0^\infty \frac{A_0^4}{[1+(f/f_B)^2]^2}\,df = f_B\int_0^\infty \frac{dx}{(1+x^2)^2} = \frac{\pi}{4}f_B.$$

7.3 For this problem see also IEEE Transactions on Circuit Theory, Vol CT-20, NO. 5, September 1973, pp. 524–532. We have

$$\int_0^\infty |H_{LP}|^2\,df = f_0\int_0^\infty \frac{d(f/f_0)}{[1-(f/f_0)^2]^2+(f/f_0)^2/Q^2}$$

$$\int_0^\infty |H_{BP}|^2\,df = \frac{f_0}{Q^2}\int_0^\infty \frac{(f/f_0)^2}{[1-(f/f_0)^2]^2+(f/f_0)^2/Q^2}\,d(f/f_0).$$

Using integral tables (see, for instance, Gradstein an Ryzhik, "Table of Integrals, Series, and Products," Academic Press, New York, 1965) it is found that

$$\int_0^\infty |H_{LP}|^2\,df = Q^2\int_0^\infty |H_{BP}|^2 = Q\frac{\pi}{2}f_0.$$

Since $|H_{BP}|_{max}=1$, it follows that $NEB_{BP} = (\frac{\pi}{2}f_0)/Q$. The higher the Q, the narrower the $|H_{BP}|$ curve, so it makes sense that NEB_{BP} decreases with Q.

(7.2)

For $Q < 1/\sqrt{2}$, $|H_{LP}|_{max} = 1$, so $NEB_{LP} = Q\frac{\pi}{2}f_0$. For higher values of Q, we can either retain this expression, or we can rigorously comply with the definition of Eq. (7.13). In the former case we observe that NEB increases with Q due to the additional area available in the frequency region where peaking occurs. In the latter we must use $NEB_{LP} = |H_{LP}|_{max}^{-1} \times Q\frac{\pi}{2}f_0$. For large Qs (say $Q \geq 5$), $|H_{LP}|_{max} \cong Q$, so $NEB_{LP} \cong (1/Q^2)Q\frac{\pi}{2}f_0 = (\frac{\pi}{2}f_0)/Q = NEB_{BP}$, indicating that for high Qs the NEBs of the two functions tend to be the same. Indeed, in this case, most of the noise comes from the frequency region of peaking, where the two functions are virtually indistinguishable.

$\boxed{7.4}$ (a) $H = 1/(1+jf/f_0)^2$, $f_0 = 1/2\pi RC$; $|H|^2 = 1/[1+(f/f_0)^2]^2$; $NEB = \int_0^\infty df/[1+(f/f_0)^2]^2 = f_0\int_0^\infty \frac{dx}{(1+x^2)^2}$. Using the integral tables, or also applying the results of Problem 7.2 with $Q=0.5$, we get $NEB = (\pi/4)f_0 = 0.785 f_0$.

(b) $H = (jf/f_0)/[1+jf/f_0]^2$; $|H|^2 = (f/f_0)^2/[1+(f/f_0)^2]^2$; $|H|^2_{max} = (1/2)^2 = 1/4$;

7.3

$$NEB = 4f_0\int_0^\infty \frac{x^2}{(1+x^2)^2}\,dx = \pi f_0 .$$

(c) The order in which the stages are cascaded is mathematically irrelevant, so $NEB = \pi f_0$.

[redacted]

7.4

$$NEB = \frac{1}{50^2}\int_0^\infty |A_n|^2 df = \frac{1}{50^2}\int_0^\infty (1+50H_{BP})^2 \frac{df}{1+(f/10^6)^2}$$

$$\cong \frac{1}{50^2}\left[\int_0^\infty |50H_{BP}|^2 df + \int_0^\infty \frac{df}{1+(f/10^6)^2}\right]$$

$$= \frac{1}{50^2}\left[50^2\frac{\pi f_0}{2Q} + 1.57 f_t\right] \cong 500 + 628 = 1.3\,\text{kHz}.$$

$\boxed{7.6}$ The noise gain, normalized to unity at dc, can be expressed as

$$|A_n(jf)|^2 = \frac{1+(f/f_1)^2}{[1+(f/f_2)^2][1+(f/f_3)^2]}.$$

At low frequencies ($f<f_1<f_3$) we can write

$$|A_n(jf)|^2 \simeq \frac{1}{1+(f/f_2)^2},$$

whose integral from 0 to f_1 can be approximated with the integral from 0 to ∞ to yield a contribution of $1.57 f_2 = 1.57\times 50 = 78.5\,\text{Hz}$. At high frequencies ($f>f_1>f_2$) we can write

$$|A_n(jf)|^2 \cong \left(\frac{f_2}{f_1}\right)^2 \frac{1}{1+(f/f_3)^2},$$

whose integral from f_1 to ∞ yields a contribution of $(f_2/f_1)^2(1.57f_3 - f_1) = (50/500)^2(1.57\times 2{,}122 - 500) = 28.3\,\text{Hz}$. Thus, $NEB \cong 78.5 + 28.3 \simeq 110\,\text{Hz}$. This is confirmed by PSpice:

(7.5)

```
Problem 7.6:
Vi 1 0 ac 1
R1 1 0 1
E2 2 0 Laplace {V(1,0)}={1+s/3142}
R2 2 0 1
E3 3 0 Laplace {V(2,0)}={1/((1+s/314.2)*(1+s/13333))}
R3 3 0 1
.ac dec 10 0.1 1Meg
.probe
.end
```



7.7

```
PROBLEM 7.7
Vi 1 0 AC 1
R1 1 0 1
E2 2 0 Laplace {V(1,0)}={(1+s/62.83)*(1+s/6283)}
R2 2 0 1
E3 3 0 Laplace {V(2,0)}={1/((1+s/628.3)*(1+s/1257))}
R3 3 0 1
E4 4 0 Laplace {V(3,0)}={1/((1+s/62832)*(1+s/62832))}
R4 3 0 1
.ac dec 10 0.1 1Meg1
.probe
.end
```

7.6

From the plots, $|A_n|_{max} = 6.72$ V/V, NEB = 1.12 kHz.



7.8



7.7

For $f \le 1\,\text{Hz}$, $e_{no}^2 = (1000\times10^{-9}\sqrt{f})^2 = 10^{-12} f$;

$E_{no1}^2 = 10^{-12}\int_0^{10} f\,df = \frac{1}{2}10^{-12} f^2\Big|_0^{10} = 5\times10^{-11}\ \text{V}^2.$

For $10\,\text{Hz} \le f \le 100\,\text{Hz}$ we have $e_{no}^2 = 10^{-10}/f$;

$E_{no2}^2 = 10^{-10}\int_{10}^{100} \frac{df}{f} = 10^{-10}\ln\left(\frac{100}{10}\right) = 2.3\times10^{-10}\ \text{V}^2.$

For $f \ge 100\,\text{Hz}$ we have white noise going through a low-pass filter with $f_0 = 1\,\text{kHz}$;

$E_{no3}^2 = (1000\times10^{-9})^2 \times (1.57\times10^3 - 100) = 1.47\times10^{-9}\ \text{V}^2.$

$E_{no} = \sqrt{5\times10^{-11} + 2.3\times10^{-10} + 1.47\times10^{-9}} = 41.83\,\mu\text{V}.$

Pink noise tangent gives $E_{no} \cong 10^3\,\text{nV}\sqrt{1.57\times10^3}$ $= 39.6\ \text{V}$, close enough to $41.83\,\mu\text{V}$.

7.9



$10^{-2}\,\text{Hz} \le f \le 1\,\text{Hz}$: $e_{no1}^2 = 10^{-6}/f\ \ \text{V}^2/\text{Hz}$;

$E_{no1}^2 = \int_{10^{-2}}^{1} e_{no1}^2\,df = 4.6\times10^{-16}\ \text{V}^2.$

$1\,\text{Hz} \le f \le 10^2\,\text{Hz}$: $e_{no2}^2 = 10^{-16}\ \text{V}^2/\text{Hz}$;

$E_{no2}^2 = 10^{-16}(100-1) = 99\times10^{-16}\ \text{V}^2.$

(7.8)

$10^2\,\text{Hz} \le f \le 10^3\,\text{Hz}$: $e_{no3}^2 = 10^{-12}/f^2\ \text{V}^2/\text{Hz}$;

$$E_{no3}^2 = 10^{-12}\int_{10^2}^{10^3} \frac{df}{f^2} = 10^{-12}\left(-\frac{1}{f}\right)\Big|_{10^2}^{10^3} = 90\times 10^{-16}\ \text{V}^2.$$

$f \geq 10^3\,\text{Hz}$: $E_{no4}^2 = 10^{-18}\left(\frac{\pi}{2}10^4 - 10^3\right) = 147\times10^{-16}\ \text{V}^2$.

$$E_{no}^2 = (4.6 + 99 + 90 + 147)\times 10^{-16} = (0.185\ \mu\text{V})^2.$$

The pink noise line touches e_{no} at $f = 100$ Hz as well as $f = 10^4$ Hz. We can thus approximate as

$$E_{no}^2 \simeq (10\,\text{nV})^2 \times \frac{\pi}{2}\times 100 + (1\,\text{nV})^2\times\frac{\pi}{2}\times 10^4$$
$$= (0.177\ \mu\text{V})^2.$$

$\boxed{7.10}$ We have $f_L = 1/60$ and $f_0 = 1/2\pi RC = 16$ Hz. Moreover, $e_n^2 = (118\ \text{nV}/\sqrt{\text{Hz}})^2(30/f + 1)$ and $e_{10\,k\Omega}^2 = 4kTR = (12.8\ \text{nV}/\sqrt{\text{Hz}})^2$. Consequently,

$$e_{ni}^2 = e_n^2 + e_{10k\Omega}^2 = 118.7^2 + 118^2\times 30/f\ \ \text{nV}/\text{Hz}$$
$$= 118.7^2(29.65/f + 1)\ \ \text{nV}/\sqrt{\text{Hz}}.$$

$$e_{no}^2 = (118.7\times10^{-9})^2(29.65/f + 1)\frac{1}{1+(f/16)^2}.$$

Piecewise integration:

$1/60 < f < 16\,\text{Hz}$; $e_{no}^2 \cong (118.7\times10^{-9})^2\times 30.36/f$; $E_{no1} = 118.7\times10^{-9}\left[\int_{1/60}^{16}(29.65/f)\,df\right]^{1/2} \cong$

(7.9) 1.7 μV.

16 Hz < f < 29.65 Hz: $e_{mo}^2 \cong (118.7\times10^{-9})^2 \times (29.65/f)\times(16/f)^2 = (10.34\times10^{-6})^2/f^3$. Now $\int_{16}^{29.65} df/f^3 = -\frac{1}{2}\frac{1}{f^2}\Big|_{16}^{29.65} = 1/722$, so $E_{mo2} \cong$ 0.38 μV.

$f > 29.65$ Hz: $e_{mo}^2 \cong (118.7\times10^{-9})^2 \times (16/f)^2 = (1.9\times10^{-6})^2/f^2$. Now $\int_{29.65}^{\infty} df/f^2 = 1/29.65$, so $E_{mo3} = 1.9\times10^{-6}/\sqrt{29.65} = 0.35\ \mu V$.

Finally, $E_{mo} = (1.7^2 + .38^2 + .35^2) = 1.78\ \mu V$ (rms), or 11.7 μV peak-to-peak.

7.11 $R = \dfrac{4kT}{i_R^2} = \dfrac{4kT}{i_D^2} = \dfrac{4kT}{2qI_D} = \dfrac{51.5\times10^{-3}}{I_D}$

For $I_D = 50\ \mu A$, $R = 1.03\ k\Omega$; for $I_D = 1\ pA$, $R = 51.5\ G\Omega$.

7.12 (a) $e_{mo}^2 = \left|R \| \frac{1}{sC}\right|^2 i_R^2 = \dfrac{R^2}{|1+sRC|^2}\dfrac{4kT}{R}$



$= \dfrac{4kTR}{1+(f/f_0)^2}$, $f_0 = \dfrac{1}{2\pi RC}$.

$E_n = \sqrt{4kTR\times(\pi/2)f_0} = \sqrt{kT/C}$.

(b)



$i_{mo}^2 = e_R^2/|R+sL|^2$

$= \dfrac{4kTR}{|R(1+sL/R)|^2} =$

$= (4kT/R)/[1+(f/f_0)^2]$,

7.10

$f_0 = 1/(2\pi L/R)$; $I_n = \sqrt{(4kT/R)(\pi/2)f_0} = \sqrt{kT/L}$.

7.13 (a) $R(k\Omega) \cong (e_{nw}/4)^2 \cong (20/4)^2 \Rightarrow R \cong 25\ k\Omega$.

(b) $I = (0.5\times10^{-12})^2/2q = 780\ nA$. This is much larger than the input bias current I_B (80 nA typical), indicating the presence of other noise sources besides shot noise due to I_B alone.

7.14



(a) 741: $f_B = \beta f_t = 10^6/11 = 91$ kHz; $E_{noe} = 11\times20\times10^{-9}[200 \ln(91\times10^3/0.1) + 1.57\times91\times10^3]^{1/2} \cong 84.0\ \mu V$; $E_{noi} = 11\times91\times10^3\sqrt{2}\times0.5\times10^{-12}\times[2\times10^3 \ln(91\times10^3/0.1) + 1.57\times91\times10^3]^{1/2} \cong 292\ \mu V$; $E_{noR} = 11\times[1.65\times10^{-20}\times2\times91\times10^3\times1.57\times91\times10^3]^{1/2} \cong 228\ \mu V$; $E_{no} = (84^2 + 292^2 + 228^2)^{1/2} \cong 380\ \mu V$.

(b) OP-27: $f_B = 8\times10^6/11 = 727$ kHz; $E_{noe} = 11\times3\times10^{-9}[2.7 \ln(727\times10^3/0.1) + 1.57\times727\times10^3]^{1/2} \cong 35.3\ \mu V$; $E_{noi} = 11\times91\times10^3\sqrt{2}\times0.4\times10^{-12}\times[140 \ln(727\times10^3/0.1) + 1.57\times727\times10^3]^{1/2} = 605\ \mu V$; $E_{noR} = 11\times[1.65\times10^{-20}\times2\times91\times10^3\times1.57\times727\times10^3]^{1/2} \cong 644\ \mu V$; $E_{no} \cong 885\ \mu V$; all rms.

In both circuits current and thermal

7.11

noise far exceed voltage noise, indicating that the resistances ought to be suitably reduced. Moreover, under the given conditions, the OP-27 circuit is noisier because of the higher f_t.

7.15



$R_p = 15 // 10 = 6\text{ k}\Omega$,
$R_n = 10 // 10 // 20 = 4\text{ k}\Omega$,
$\beta = 5/25 = 0.2$ V/V.
$f_B = 0.2 \times 10^6 = 200\text{ kHz}$;
$NEB = 1.57 f_B = 314\text{ kHz}$;
$A_n = 1/\beta = 5$ V/V.

$$E_{no} = 5\left[(20\times10^{-9})^2(200 \ln 200{,}000 + 314{,}000) + (6{,}000^2 + 4{,}000^2)(0.5\times10^{-12})^2(2{,}000 \times \ln 200{,}000 + 314{,}000) + 1.65\times10^{-20} \times 10{,}000 \times 314{,}000\right]^{1/2} = 67.7\ \mu\text{V rms}.$$

7.16 OA_1: $(A-1)R = 100\text{ k}\Omega$; $R_p = 0$; $R_n = 100 // 100 = 50\text{ k}\Omega$; $\beta = 0.5$; $A_n = 2$; $f_B = 500\text{ kHz}$; $NEB = 1.57 \times 500 = 785\text{ kHz}$.

$$E_{no1} = 2\left[(20\times10^{-9})^2(200 \ln 500{,}000 + 785{,}000) + (50{,}000 \times 0.5\times10^{-12})^2(2{,}000 \times \ln 500{,}000 + 785{,}000) + 1.65\times10^{-20} \times 50{,}000 \times 785{,}000\right]^{1/2} = 76.7\ \mu\text{V rms}$$

(7.12)

OA_2: $AR = 200\ k\Omega$; $R_p = 0$; $R_n = 200 \| 100 = 67\ k\Omega$; $b = 1/3$; $A_n = 3$; $f_B = 333\ kHz$, $NEB = 523\ k\Omega$. $E_{no2} = 112.4\ \mu V$ rms.
$E_{no} = (76.7^2 + 112.4)^{1/2} = 136\ \mu V$ rms.

$\boxed{7.17}$ (a) $R_p = 0$, $R_n = 10 + 2 \| 18 = 11.8\ k\Omega$; $\beta = 2/(2+18) = 0.1\ V/V$; $f_B = 0.1 \times 8 = 800\ kHz$; $A_{no} = 1/\beta = 10\ V/V$.



$E_{noe} = 10 \times 3 \times 10^{-9} [2.7 \ln (8 \times 10^5/0.1) + 1.57 \times 8 \times 10^5]^{1/2} = 34\ \mu V$; $E_{noi} = 10 \times 11.8 \times 10^3 \times 0.4 \times 10^{-12} \times [140 \times \ln (8 \times 10^6) + 1.57 \times 8 \times 10^5]^{1/2} = 53\ \mu V$; $e_{noR} = 10\sqrt{1.65 \times 10^{-20} \times 11.8 \times 10^3 \times 1.57 \times 8 \times 10^5} = 156\ \mu V$, $E_{no} = (34^2 + 53^2 + 156^2)^{1/2} \cong 168\ \mu V$.

(b) $V_{i(rms)} = (10\ \mu A)/\sqrt{3} = 5.77\ \mu A$; $I_{ni} = E_{no}/R_{eq} = (168\ \mu V)/[10^4(1 + 18/2 + 18/10)] = 1.34\ nA$, $SNR = 20 \log_{10} [5.77 \times 10^{-6}/(1.34 \times 10^{-9})] = 72.7\ dB$.

$\boxed{7.18}$ (a) $A_{ideal} = -8\ V/V$; $\beta = 1/13\ V/V$; $f_B = \beta f_t \cong 77\ kHz$; $R_n = 10 \| \{10 + 10 \| [10 + (10 \| 10)]\} = 6.15\ k\Omega$. $E_{noe} = 13 \times 20 \times 10^{-9} [200 \ln (77 \times 10^3/0.1) + 1.57 \times 77 \times 10^3]^{1/2} = 91.4\ \mu V$; $E_{noi} = 13 \times 6.15 \times 10^3 \times 0.5 \times 10^{-12} [2 \times 10^3 \ln (77 \times 10^3/0.1) + 1.57 \times 77 \times 10^3]^{1/2} = 15.4\ \mu V$; $E_{noR} = 13(1.65 \times 10^{-20} \times 6.15 \times$

(7.13)

$10^3 \times 1.57 \times 77 \times 10^3]^{1/2} = 45.5\,\mu V$. $E_{no} = (91.4^2 + 15.4^2 + 45.5^2)^{1/2} \cong 103\,\mu V$.

(b) $V_{i(rms)} = (0.5^2 + 0.25^2)^{1/2} = 0.56$ V; $E_{ni} = 103/13 = 7.92\,\mu V$; $SNR = 20\log_{10}(0.5/7.92\times 10^{-6}) = 97$ dB.

$\boxed{7.19}$ Ignore i_n. Noise model is as follows:



$$\frac{1}{\beta} = 1 + \frac{1/j\omega C}{R} = \frac{1 + j(f/f_0)}{j(f/f_0)},$$

$$f_0 = \frac{1}{2\pi RC} = 1\text{KHz}.$$



The contribution from e_n to e_{no} is $e_{no1} = |A_n|\, e_n$. The pink noise tangent principle shows that most noise comes from below f_{ce} and from near f_t. Thus, for $f_L < f < f_{ce}$ we can write

$$e_{no1} \cong 18\times 10^{-9}\left(\frac{200}{f}\right)^{1/2} \times \frac{1}{f/10^3} = 2.55\times 10^{-4}/f^{3/2}.$$

7.14

Hence, $E_{no1} \simeq 2.55\times10^{-4}\left[\int_1^{200}\frac{1}{f^3}df\right]^{1/2} =$
$2.55\times10^{-4}\left[\frac{1}{2}\left(\frac{1}{1^2}-\frac{1}{200^2}\right)\right]^{1/2} = 180\,\mu V.$
For $f > f_0$ we have $E_{no2} \simeq 18\times10^{-9}\times(1.57\times3\times10^6)^{1/2} = 39\,\mu V.$

The contribution from e_R to e_{no} is, up to f_t, $e_{no2} = (kTR)^{1/2}\times1/(f/f_0) =$
$(1.65\times10^{-20}\times158\times10^3)^{1/2}\times10^3/f = 5.1\times10^{-5}/f.$
Past f_t the contribution rolls off quadratically and can therefore be ignored. Thus,
$E_{no3} \simeq 5.1\times10^{-5}\left[\int_1^{3\times10^6}\frac{1}{f^2}df\right]^{1/2} \simeq 51\,\mu V.$
Finally, $E_{no} = (E_{no1}^2 + E_{no2}^2 + E_{no3}^2)^{1/2} =$
$(180^2 + 39^2 + 51^2)^{1/2} = 191\,\mu V$ rms.

$\boxed{7.20}$ In the single-op-amp realization, e_n is magnified by 10^3 with NEB = 1.57×1 KHz. In the two-op-amp cascade realization, the first-stage e_n is magnified by 10^3 with NEB $= 1.11\times20.35$ KHz, as seen in Example 6.2, and additional noise is produced by the second stage. Thus, the two-op-amp realization is at least $[(1.11\times20.35)/(1.57\times1)]^{1/2} = 3.8$ times as noisy as the single-op-amp realization.

$\boxed{7.21}$ $A_0 = 10^3$ V/V $\Rightarrow R_2 = (10^3-1)R_1 \cong 10^3 R_1$; $R_n = R_1 \| R_2 \cong R_1$



$\beta_1 \cong 1$ V/V

$\beta_2 \cong 10^{-3}$ V/V

$f_{B1} \cong f_t =$ 8 MHz; $f_{B2} = \beta_2 f_t = 8$ kHz. Since $f_{B1} \gg f_{B2}$, we can write

$$e_{noe}^2 \cong \left[\left(1+\frac{R_2}{R_1}\right)^2 e_{n2}^2 + \frac{R_2^2}{R_1^2}\left(1+\frac{R_1}{R_2}\right)^2 e_{n1}^2\right]\frac{1}{1+(f/f_{B2})^2}$$

$$\cong 2e_n^2 \frac{(10^3)^2}{1+(f/8\times 10^3)^2}$$

$$\Rightarrow E_{noe} = 10^3\sqrt{2}\times 3\times 10^{-9}\left[2.7 \ln\frac{8\times 10^3}{0.1} + 1.57\times 8\times 10^3\right]^{1/2}$$

$$= 476\ \mu\text{V}.$$

Likewise, the contributions from current and resistor noise are

$E_{noi}^2 + E_{noR}^2 = 2\times 10^6\{R_1^2\times(0.4\times 10^{-12})^2[140 \ln(8\times 10^4) + 1.57\times 8\times 10^3] + R_1[1.65\times 10^{-20} + 1.57\times 8\times 10^3]\}$ $= 4.52\times 10^{-15} R_1^2 + 4.14\times 10^{-10} R_1$. Imposing $E_{noi}^2 + E_{noR}^2 \le E_{noe}^2/3$ gives $R_1 \le 180\ \Omega$. Use $R_1 = 178\ \Omega$, $R_2 = 178$ kΩ. Then, $E_{no} \cong 550\ \mu$V.

(7.16)

7.22 (a)



Let $A_I = 1+2R_3/R_G$. The contribution to e_{no}^2 from e_{n1} and e_{n2} is $A_I^2(e_{n1}^2+e_{n2}^2) = 2A_I^2 e_n^2$. The contribution from i_{n1} and i_{n2} is $R_3^2 i_{n1}^2 + R_3^2 i_{n2}^2 = 2R_3^2 i_n^2 = (2R_3)^2(i_n^2/2) = A_I^2[(2R_3)^2/A_I^2]\, i_n^2/2 = A_I^2[R_G \| (2R_3)]^2 \frac{i_n^2}{2}$. The contribution from i_{R_G}, e_{R_3}, and e_{R_3} is $(R_3+R_3)^2 i_{R_G}^2 + e_{R_3}^2 + e_{R_3}^2 = (2R_3)^2\frac{4kT}{R_G} + 2\times 4kTR_3 = 4kT(2R_3)(1+2R_3/R_G) = A_I \times 4kT(2R_3) = A_I^2 \times 4kT(2R_3/A_I) = A_I^2 \times 4kT[R_G \| (2R_3)]$. So,

$$e_{ni}^2 = e_{no}^2/A_I^2 = 2e_n^2 + [R_G\|(2R_3)]^2 i_n^2/2 + 4kT[R_G\|(2R_3)].$$

(b) $f_{BI} = \beta f_t = 8\times 10^6/10^3 = 8$ kHz; $R_G\|(2R_3) \cong R_G \simeq 100\ \Omega$. $E_{noe} = 10^3 \times \sqrt{2} \times 3\times 10^{-9} \times$

(7.17) $[2.7\ln(8\times10^3/0.1)+1.57\times8\times10^3]^{1/2} = 476\ \mu V$; $E_{noi} = 10^3\times10^2\times(0.4\times10^{-12}/\sqrt{2})\times[140\ln(8\times10^4)+1.57\times8\times10^3]^{1/2} = 3.4\ \mu V$; $E_{noR} = 10^3[1.65\times10^{-20}\times10^2\times1.57\times8\times10^3]^{1/2} = 144\ \mu V$. $E_{no} = (476^2+3.4^2+144^2)^{1/2} = 497\ \mu V$.

$\boxed{7.23}$ (a) $\beta_I = 1/2000$ V/V, $\beta_{II} = 2/3$ V/V; $f_{BI} = \beta_I f_t = 8\times10^6/2000 = 4$ kHz; $f_{BII} = 5.33$ MHz. Since $f_{BII} \gg f_{BI}$, the total output noise of the first stage is transmitted to the output with a gain of 0.5 V/V, thus contributing the rms noise $0.5\times2000\{2(3\times10^{-9})^2[2.7\ln(4\times10^3/0.1)+1.57\times4\times10^3]+[(50\times0.4\times10^{-12})^2/2]\times[140\ln(4\times10^4)+1.57\times4\times10^3]+1.65\times10^{-20}\times50\times1.57\times4\times10^3\}^{1/2} = 345\ \mu V$. The second stage contributes the rms noise $1.5\{(3\times10^{-9})^2[2.7\ln(5.33\times10^6/0.1)+1.57\times5.33\times10^6)+[(50\|100)^2\times(10^3)^2\times(0.4\times10^{-12})^2\times[140\ln(5.33\times10^7)+1.57\times5.33\times10^6]+1.65\times10^{-20}\times2\times(50\|100)\times10^3\times1.57\times5.33\times10^6\}^{1/2} = 166\ \mu V$. Finally, $E_{no} = (345^2+166^2)^{1/2} = 383\ \mu V$.

(b) $E_{ni} = (383\ \mu V)/10^3 = 383$ nV; SNR $= 20\log_{10}[(10\times10^{-3}/\sqrt{2})/(383\times10^{-9})] = 85.3$ dB.

(7.18)

7.24




```
Problem 7.24:
.subckt noisyCFA vP vN vO
*Input noise sources:
IDe 0 11 dc 3.12uA
De 11 0 De
*fce = 50kHz
.model De D (KF=1.6E-14,AF=1)
Ce 11 12 1GF
vse 12 0 dc 0V
*enw = 2.4 nV/sqrt(Hz)
he 1 vP vse 2.4k
IDp 0 21 dc 3.12uA
Dp 21 0 Dp
*fcip = 100kHz
.model Dp D (KF=3.2E-14,AF=1)
Cp 21 22 1GF
vsp 22 0 dc 0V
*inpw = 3.8 pA/sqrt(Hz)
fp 0 1 vsp 3.8
IDn 0 31 dc 3.12uA
Dn 31 0 Dn
*fcin = 100 kHz
.model Dn D (KF=3.2E-14,AF=1)
Cn 31 32 1GF
vsn 32 0 dc 0V
*inw = 20 pA/sqrt(Hz)
fn 0 vN vsn 20
*Noiseless CFA:
*z0 = 710 k, fb = 350 kHz, rn = 50 Ohm
ein 100 0 vp 0 1 ;input buffer
rn 100 200 50 ;buffer's output resistance
vs 200 vN dc 0 ;0-V source to sense iN
fCFA 0 300 vs 1 ;CCCS
Req 300 0 710k ;dc gain
Ceq 300 0 0.641pF ;fa=350kHz
eout vO 0 300 0 1 ;output buffer
.ends noisyCFA
*Main circuit:
Vi 1000 0 ac 1V
Rs 1000 1 100
R1 0 2 166.7
R2 2 3 1.5k
XCFA 1 2 3 noisyCFA
.ac dec 10 10KHz 10GHz
.noise v(3) vi 10
.probe
.end
```

7.19



$$e_{no} \cong 38\times10^{-9}\times(1.57\times 120\times10^{6})^{1/2} \cong 522\ \mu V.$$

$\boxed{7.25}$ $1/\beta = 10^3$ V/V; $R_n \cong R_p = 10\,\Omega \cong 0$.
$0.120\times10^{-3} = 10^3 e_n\sqrt{100} \Rightarrow e_n = 12\ \mathrm{nV}/\sqrt{\mathrm{Hz}}$.
With the 500-kΩ resistors in place, $R_p \cong R_n = 500$ kΩ; $2.25\times10^{-3} = 10^3\{(12\times10^{-9})^2\times100 + 2\times(500\times10^3 i_n)^2\times100 + 4kT(2\times500\times10^3)100\}^{1/2} \Rightarrow i_n = 2.6\ \mathrm{pA}/\sqrt{\mathrm{Hz}}$.

$\boxed{7.26}$ (a) Denoting the voltage at the op amp output pin as V_1, we have, by KVL at V_n and V_o,

$$\frac{V_o-V_i}{mR} = I_i + \frac{V_i-V_1}{1/smC}, \quad \frac{V_1-V_o}{R} = \frac{V_o}{1/sC} + \frac{V_o-V_i}{mR}.$$

Eliminating V_1 gives

$$V_o = \frac{mRI_i + [1+s(m+1)nRC]V_i}{1+s(m+1)nRC + s^2 mnR^2C^2}$$

$$= H_{LP}\,mRI_i + (H_{LP}+H_{BP})V_i.$$

7.20

$f_0 = 1/[2\pi\sqrt{mn}\,RC]$, $Q = \sqrt{m/n}/(m+1)$.

(b)



7.27 $n=1$; $Q = 1/2 \Rightarrow m=1$; $R = 1/(2\pi\times 100 \times 0.1\times 10^{-6}) = 15.9\ k\Omega$.

```
Problem 7.27:
.subckt noisyOA vP vN vO
*enw = 20 nV/sqrt(Hx), fce = 200 Hz:
IDe 0 11 dc 3.12uA
De 11 0 De
.model De D (KF=6.41E-17,AF=1)
Ce 11 12 1GF
vse 12 0 dc 0V
he 1 vP vse 20k
*inpw = 0.5 pA/sqrt(Hz), fcip = 2 kHz:
IDp 0 21 dc 3.12uA
Dp 21 0 Dp
.model Dp D (KF=6.41E-16,AF=1)
Cp 21 22 1GF
vsp 22 0 dc 0V
fp 0 1 vsp 0.5
*innw = 0.5 pA/sqrt(Hz), fcin = 2 kHz:
IDn 0 31 dc 3.12uA
Dn 31 0 Dn
.model Dn D (KF=6.41E-16,AF=1)
Cn 31 32 1GF
vsn 32 0 dc 0V
fn 0 vN vsn 0.5
*Noiseless op amp (a0 = 200V/mV, fb = 5 Hz):
ea0 5 0 1 vN 200k
Req 5 6 1
Ceq 6 0 31.83mF
ebuf vO 0 6 0 1
.ends noisyOA
*Main circuit:
Vi 1 0 ac 1V
Ri 1 0 1
RF 2 4 15.9k
CF 2 3 0.1uf
R 3 4 15.9k
C 4 0 0.1uF
XOA 1 2 3 noisyOA
.ac dec 10 0.01Hz 1MegHz
.noise v(4) vi 10
.probe
.end
```

7.21


10uV
10nV
10pV
e_{no}
□ v(onoise)
1.5u
1.0u
0.5u
0
1.49 μV
10mHz
10Hz
10KHz
1.0MHz
□ sqrt(s(v(onoise)*v(onoise)))
Frequency

7.28


RF

7.22

```
Problem 7.28:
.subckt noisyOA vP vN vO
*enw = 3 nV/sqrt(Hx), fce = 2.7 Hz:
IDe 0 11 dc 3.12uA
De 11 0 De
.model De D (KF=8.6E-19,AF=1)
Ce 11 12 1GF
vse 12 0 dc 0V
he 1 vP vse 3k
*inpw = 0.4 pA/sqrt(Hz), fcip = 140 Hz:
IDp 0 21 dc 3.12uA
Dp 21 0 Dp
.model Dp D (KF=4.5E-17,AF=1)
Cp 21 22 1GF
vsp 22 0 dc 0V
fp 0 1 vsp 0.4
*innw = 0.4 pA/sqrt(Hz), fcin = 140 Hz:
IDn 0 31 dc 3.12uA
Dn 31 0 Dn
.model Dn D (KF=4.5E-17,AF=1)
Cn 31 32 1GF
vsn 32 0 dc 0V
fn 0 vN vsn 0.4
*Noiseless op amp (a0 = 8V/uV, fb = 1 Hz):
ea0 5 0 1 vN 8Meg
Req 5 6 1
Ceq 6 0 159mF
ebuf vO 0 6 0 1
.ends noisyOA
*Main circuit:
*LT009 noise model
*enw = 118 nV/sqrt(Hx), fce = 30 Hz:
IDz 0 41 dc 3.12uA
Dz 41 0 Dz
.model Dz D (KF=4.8E-18,AF=1)
Cz 41 42 1GF
vsz 42 0 dc 0V
hz 100 0 vsz 118k
Rz 100 0 1
R3 100 200 15.9k
C3 200 0 10uF
Vi 1 200 ac 1V
Ri 1 200 1
RF 2 4 15.9k
CF 2 3 1uf
R 3 4 15.9k
C 4 0 1uF
XOA 1 2 3 noisyOA
.ac dec 10 0.01Hz 1MegHz
.noise v(4) vi 10
.probe
.end
```

7.23



7.29 (a)



$C = 1/(2\pi \times 10^3 \times 200\times 10^3) =$ 795 pF. It is easily seen that

$$\frac{1}{\beta} = \frac{1}{\beta_0}\,\frac{1+j(f/f_z)}{1+j(f/f_p)},$$

$\frac{1}{\beta_0} = 1 + \frac{R_2}{R_1} = 3$, $f_p = \frac{1}{2\pi R_2 C} = 1\text{ kHz}$, $f_z = \frac{1}{2\pi R_1 /\!/ R_2 C} =$ 3 kHz; $\beta_\infty = 1$ V/V; $\beta_\infty f_t = 1$ MHz.

We identify two types of noise:

<u>1. Noninverting-input noise,</u> consisting of e_n, e_{R_3}, and $R_3 i_n$. This noise is amplified by A_n, where $A_n \cong 1/\beta$ for $f < 1$ MHz, and $A_n \cong a$ for $f > 1$ MHz. The pink noise tangent indicates that most noise comes from the vicinity of 1 MHz, where $|A_n| \cong 1$ V/V. Thus,

$$E_{no1} = [(e_{nw}^2 + R_3^2 i_{nW}^2 + 4kTR_3)\times 1.57\times 10^6]^{1/2} =$$

7.24

$[(20\times10^{-9})^2 + (68\times10^3\times0.5\times10^{-12})^2 + 1.65\times10^{-20}\times 68\times10^3]^{1/2} \times (1.57\times10^6)^{1/2} = 64.8\,\mu V.$

2. Inverting-input noise, consisting of i_{R_1}, i_{R_2}, and i_n flowing through

$$Z_2 = \frac{R_2}{1+j(f/f_p)} = \frac{200\,k\Omega}{1+j(f/10^3)}.$$

Denoting the density of this noise as i_2, we have $i_2^2 = i_n^2 + i_{R_1}^2 + i_{R_2}^2 = i_n^2 + 4KT/(R_1 /\!/ R_2)$, where $R_1 /\!/ R_2 = 100/\!/200 = 67\,k\Omega$. Thus,

$$i_2^2 = (0.5\times10^{-12})^2\left(\frac{2000}{f}+1\right) + \frac{1.65\times10^{-20}}{67\times10^3}$$

$$\simeq (0.7\times10^{-12})\left(\frac{10^3}{f}+1\right).$$

The corresponding output power density is

$$e_{no2}^2 = |Z_2|^2 i_2^2 = (200\times10^3\times0.7\times10^{-12})\times \left(\frac{10^3}{f}+1\right)\frac{1}{1+(f/10^3)^2}$$

$$= (140\,mV)^2\left(\frac{10^3}{f}+1\right)\frac{1}{1+(f/10^3)^2}.$$

For $f < 10^3$ Hz, $e_{no2}^2 \simeq (140\,mV)^2\,\frac{10^3}{f}$, so that $E_{no2(1)} = (140\,mV)\times[10^3 \ln(10^3/0.01)]^{1/2} = 15\,\mu V.$

For $f > 10^3$ Hz, $e_{no2}^2 \simeq (140\,mV)^2\left(\frac{10^3}{f}\right)^2$, so that

7.25

$E_{no2}(2) = (140\,nV) \times \left[\int_{10^3}^{\infty} \frac{10^6}{f^2} df\right]^{1/2} = 140 \times \left\{10^6 \left[-\frac{1}{f}\right]_{10^3}^{\infty}\right\}^{1/2} = 4.43\,\mu V$. Thus, $E_{no2} = (15^2 + 4.43^2)^{1/2} \simeq 16\,\mu V$. Finally, $E_{no} = (E_{no1}^2 + E_{no2}^2)^{1/2} = (64.8^2 + 16^2)^{1/2} = 67\,\mu V$.

Using the capacitor reduces E_{no} from $154\,\mu V$ to $67\,\mu V$; however, it also reduces the signal gain bandwidth from 333 KHz to 1 KHz.

(b) Let $f_3 = 1/(2\pi \times 68 \times 10^3 \times 0.1 \times 10^{-6}) = 23\,Hz$. Such a low corner frequency renders the effect of i_{np} and i_{R_3} insignificant, so that we now have $E_{no1} \simeq e_n \sqrt{1.57 f_c} = 20\,nV \sqrt{1.57 \times 10^6} = 25\,\mu V$. Then, $E_{no} = (25^2 + 16^2)^{1/2} = 29.7\,\mu V$, which is less than half the noise without the $0.1\,\mu F$ cap.

The above results are confirmed by PSpice:



7.26

7.30

```
Problem 7.30 (enoe)
.subckt opa vP vN vO
*enw = 4.5 nV/sqrt(Hz), fce = 100 Hz
IDe 0 11 dc 3.12uA
De 11 0 De
.model De D (KF=3.204E-17,AF=1)
Ce 11 12 1GF
vse 12 0 dc 0V
he 1 vP vse 4.5k
Rhe 1 vP 1G ;avoids floating nodes
*innw = 0.566 fA/sqrt(Hz), fcin = 0
IDi 0 21 dc 1pA
Di 21 0 Di
.model Di D (KF=0)
Ci 21 22 1GF
vsi 22 0 dc 0V
fi 0 vN vsi 1u;***zero innw
*Noiseless op amp:
*a0 = 1V/uV, fb = 16 Hz, ft = 16 MHz
ea0 2 0 1 vN 1Meg
Req 2 3 1
Ceq 3 0 9.947mF
ebuf vO 0 3 0 1
.ends opa
*Main circuit:
Ii 2 1 ac 1uA
Vs 1 0 dc 0
R1 2 0 100G
C1 2 0 45pF
G2 2 3 2 3 1E-7;***noiseless R2
C2 2 3 0.5pF
X1 0 2 3 opa
.ac dec 10 0.1Hz 1GHz
.noise V(3) Vs 10
.probe
.end
Problem 7.30 (enoi)
.subckt opa vP vN vO
*enw = 4.5 nV/sqrt(Hz), fce = 100 Hz
IDe 0 11 dc 3.12uA
De 11 0 De
.model De D (KF=3.204E-17,AF=1)
Ce 11 12 1GF
vse 12 0 dc 0V
he 1 vP vse 1u;***zero enw
Rhe 1 vP 1G ;avoids floating nodes
*innw = 0.566 fA/sqrt(Hz), fcin = 0
IDi 0 21 dc 1pA
Di 21 0 Di
.model Di D (KF=0)
Ci 21 22 1GF
vsi 22 0 dc 0V
fi 0 vN vsi 1
*Noiseless op amp:
*a0 = 1V/uV, fb = 16 Hz, ft = 16 MHz
```

7.27

```
ea0 2 0 1 vN 1Meg
Req 2 3 1
Ceq 3 0 9.947mF
ebuf vO 0 3 0 1
.ends opa
*Main circuit:
Ii 2 1 ac 1uA
Vs 1 0 dc 0
R1 2 0 100G
C1 2 0 45pF
G2 2 3 2 3 1E-7;***noiseless R2
C2 2 3 0.5pF
X1 0 2 3 opa
.ac dec 10 0.1Hz 1GHz
.noise V(3) Vs 10
.probe
.end
Problem 7.30 (enoR)
.subckt opa vP vN vO
*enw = 4.5 nV/sqrt(Hz), fce = 100 Hz
IDe 0 11 dc 3.12uA
De 11 0 De
.model De D (KF=3.204E-17,AF=1)
Ce 11 12 1GF
vse 12 0 dc 0V
he 1 vP vse 1u;***zero enw
Rhe 1 vP 1G ;avoids floating nodes
*innw = 0.566 fA/sqrt(Hz), fcin = 0
IDi 0 21 dc 1pA
Di 21 0 Di
.model Di D (KF=0)
Ci 21 22 1GF
vsi 22 0 dc 0V
fi 0 vN vsi 1u;***zero innw
*Noiseless op amp:
*a0 = 1V/uV, fb = 16 Hz, ft = 16 MHz
ea0 2 0 1 vN 1Meg
Req 2 3 1
Ceq 3 0 9.947mF
ebuf vO 0 3 0 1
.ends opa
*Main circuit:
Ii 2 1 ac 1uA
Vs 1 0 dc 0
R1 2 0 100G
C1 2 0 45pF
R2 2 3 10Meg
C2 2 3 0.5pF
X1 0 2 3 opa
.ac dec 10 0.1Hz 1GHz
.noise V(3) Vs 10
.probe
.end
Problem 7.30 (eno)
.subckt opa vP vN vO
*enw = 4.5 nV/sqrt(Hz), fce = 100 Hz
IDe 0 11 dc 3.12uA
De 11 0 De
.model De D (KF=3.204E-17,AF=1)
Ce 11 12 1GF
```

(7.28)

```
vse 12 0 dc 0V
he 1 vP vse 4.5k
Rhe 1 vP 1G ;avoids floating nodes
*innw = 0.566 fA/sqrt(Hz), fcin = 0
IDi 0 21 dc 1pA
Di 21 0 Di
.model Di D (KF=0)
Ci 21 22 1GF
vsi 22 0 dc 0V
fi 0 vN vsi 1
*Noiseless op amp
*a0 = 1V/uV, fb = 16 Hz, ft = 16 MHz
ea0 2 0 1 vN 1Meg
Req 2 3 1
Ceq 3 0 9.947mF
ebuf vO 0 3 0 1
.ends opa
*Main circuit:
Ii 2 1 ac 1uA
Vs 1 0 dc 0
R1 2 0 100G
C1 2 0 45pF
R2 2 3 10Meg
C2 2 3 0.5pF
X1 0 2 3 opa
.ac dec 10 0.1Hz 1GHz
.noise V(3) Vs 10
.probe
.end
```



7.29



7.31 Adding a 2-pF capacitance in parallel with R_2 increase C_2 by a factor of 5. The main effect is to lower f_p from 31.8 kHz to 31.8/5 = 6.3 kHz, lower $1/\beta_\infty$ from 91 V/V to 19.2 V/V, and rise f_x from 176 kHz to 833 kHz. We can estimate the new total noise as

$$E_{no(new)} \cong E_{no(old)} \frac{19.2}{91} \sqrt{833/176} \cong 222/2.18$$

= 102 μV. These results are confirmed by the accompanying PSpice program, which indicates a bandwidth reduction from 35 kHz to 6.3 kHz, and a noise reduction from 230 μV to 107 μV.

(7.30)

```
Problem 7.31 (with C2 = 2.5 pF)
.subckt opa vP vN vO
*enw = 4.5 nV/sqrt(Hz), fce = 100 Hz
IDe 0 11 dc 3.12uA
De 11 0 De
.model De D (KF=3.204E-17,AF=1)
Ce 11 12 1GF
vse 12 0 dc 0V
he 1 vP vse 4.5k
Rhe 1 vP 1G ;avoids floating nodes
*innw = 0.566 fA/sqrt(Hz), fcin = 0
IDi 0 21 dc 1pA
Di 21 0 Di
.model Di D (KF=0)
Ci 21 22 1GF
vsi 22 0 dc 0V
fi 0 vN vsi 1
*Noiseless op amp
*a0 = 1V/uV, fb = 16 Hz, ft = 16 MHz
ea0 2 0 1 vN 1Meg
Req 2 3 1
Ceq 3 0 9.947mF
ebuf vO 0 3 0 1
.ends opa
*Main circuit:
Ii 2 1 ac 1uA
Vs 1 0 dc 0
R1 2 0 100G
C1 2 0 45pF
R2 2 3 10Meg
C2 2 3 2.5pF
X1 0 2 3 opa
.ac dec 10 0.1Hz 1GHz
.noise V(3) Vs 10
.probe
.end
```

7.31



7.32 The accompanying PSpice program and displays indicate that adding a filter with $C_c = 0.5\,pF$, $R_3 = 1\,k\Omega$, and $C_3 = 10\,nF$, lowers noise from 230 μV to 80 μV, while the half-power bandwidth is lowered from 35 kHz to 24 kHz.

7.32

```
Problem 7.32
.subckt opa vP vN vO
*enw = 4.5 nV/sqrt(Hz), fce = 100 Hz
IDe 0 11 dc 3.12uA
De 11 0 De
.model De D (KF=3.204E-17,AF=1)
Ce 11 12 1GF
vse 12 0 dc 0V
he 1 vP vse 4.5k
Rhe 1 vP 1G
*innw = 0.566 fA/sqrt(Hz), fcin = 0
IDi 0 21 dc 1pA
Di 21 0 Di
.model Di D (KF=0)
Ci 21 22 1GF
vsi 22 0 dc 0V
fi 0 vN vsi 1
*Noiseless op amp
*a0 = 1V/uV, fb = 16 Hz, ft = 16 MHz
ea0 2 0 1 vN 1Meg
Req 2 3 1
Ceq 3 0 9.947mF
ebuf vO 0 3 0 1
.ends opa
*Main circuit:
Ii 2 1 ac 1uA
Vs 1 0 dc 0
R1 2 0 100G
C1 2 0 45pF
R2 2 3 10Meg
C2 2 3 0.5pF
Cc 2 4 0.5pF
R3 3 4 1k
C3 3 0 10nF
X1 0 2 4 opa
.ac dec 10 0.1Hz 1GHz
.noise V(3) Vs 10
.probe
.end
```



7.33



$\boxed{7.33}$



Let $(R_3 \| R_4) \ll R_2$, so that $R_{eq} = KR_2$, $K = 1 + R_4/R_3$.

$\beta = (V_n/V_o)_{I_i = 0}$. Applying KCL at node V_n,

$$\frac{V_o/K - V_n}{R_2} + \frac{V_o - V_n}{1/sC_2} = \frac{V_n}{R_1/(1+sR_1C_1)} + I_i \Rightarrow$$

7.34

$$V_o(1+sKR_2C_2) = KV_n\left[1+sR_2(C_1+C_2)+\frac{R_2}{R_1}\right]+KR_2I_i.$$

$$\frac{1}{\beta} = \frac{V_o}{V_i}\bigg|_{I_i=0} = \left(1+\frac{R_2}{R_1}\right)\left(1+\frac{R_4}{R_3}\right)\frac{1+jf/f_z}{1+jf/f_p},$$

$$f_z = \frac{1}{2\pi(R_1 /\!/ R_2)(C_1+C_2)},\quad f_p = \frac{1}{2\pi(1+R_4/R_3)R_2C_2};$$

$$A_s(\text{ideal}) = \frac{V_o}{I_i}\bigg|_{V_n=0} = \frac{(1+R_4/R_3)R_2}{1+jf/f_p};\quad f_x = \frac{f_t}{1+C_1/C_2}$$

$$A_n = \frac{1}{\beta}\frac{1}{1+jf/f_x};\quad A_s = A_s(\text{ideal})\frac{1}{1+jf/f_x}.$$

7.34 $1/\beta_0 \cong 1+R_4/R_3 = 10$ V/V; $1/\beta_\infty = 1+C_1/C_2 = 91$ V/V; $f_z = 3.5$ kHz; $f_p = 32$ kHz; $f_x = \beta_\infty f_t = 176$ kHz. $E_{noe} \cong 91\times 4.5\times10^{-9}[1.57\times(176-3.5)10^3]^{1/2} = 213\ \mu V$; $E_{noR} = [(1+R_4/R_3)kT/C_2]^{1/2} = 287\ \mu V$; $E_{no} \cong (213^2+287^2)^{1/2} = 357\ \mu V$. This is confirmed by PSpice.

7.35

```
Problem 7.34
.subckt opa vP vN vO
*enw = 4.5 nV/sqrt(Hz), fce = 100 Hz
IDe 0 11 dc 3.12uA
De 11 0 De
.model De D (KF=3.204E-17,AF=1)
Ce 11 12 1GF
vse 12 0 dc 0V
he 1 vP vse 4.5k
Rhe 1 vP 1G
*innw = 0.566 fA/sqrt(Hz)
IDi 0 21 dc 1pA
Di 21 0 Di
.model Di D (KF=0)
Ci 21 22 1GF
vsi 22 0 dc 0V
fi 0 vN vsi 1
*Noiseless op amp:
*a0 = 1V/uV, fb = 16 Hz, ft = 16 MHz
ea0 2 0 1 vN 1Meg
Req 2 3 1
Ceq 3 0 9.947mF
ebuf vO 0 3 0 1
.ends opa
*Main circuit:
Ii 2 1 ac 1uA
Vs 1 0 dc 0
R1 2 0 100G
C1 2 0 45pF
R2 2 4 1Meg
C2 2 3 0.5pF
R3 4 0 2k
R4 4 3 18k
X1 0 2 3 opa
.ac dec 10 0.1Hz 1GHz
.noise V(3) Vs 10
.probe
.end
```

7.36



7.35

```
Problem 7.35
.subckt opa vP vN vO
*enw = 4.5 nV/sqrt(Hz), fce = 100 Hz
IDe 0 11 dc 3.12uA
De 11 0 De
.model De D (KF=3.204E-17,AF=1)
Ce 11 12 1GF
vse 12 0 dc 0V
he 1 vP vse 4.5k
Rhe 1 vP 1G
*innw = 0.566 fA/sqrt(Hz)
IDi 0 21 dc 1pA
Di 21 0 Di
.model Di D (KF=0)
Ci 21 22 1GF
vsi 22 0 dc 0V
fi 0 vN vsi 1
*Noiseless op amp:
*a0 = 1V/uV, fb = 16 Hz, ft = 16 MHz
ea0 2 0 1 vN 1Meg
```

```
Req 2 3 1
Ceq 3 0 9.947mF
ebuf vO 0 3 0 1
.ends opa
*Main circuit:
Ii 2 1 ac 1nA
Vs 1 0 dc 0
R1 2 0 100G
C1 2 0 2nF
R2 2 4 36.5Meg
C2 2 3 0.5pF
R3 4 0 1k
R4 4 3 26.7k
X1 0 2 3 opa
.ac dec 10 0.1Hz 1GHz
.noise V(3) Vs 10
.probe
.end
Problem 7.35 (enoe)
.subckt opa vP vN vO
*enw = 4.5 nV/sqrt(Hz), fce = 100 Hz
IDe 0 11 dc 3.12uA
De 11 0 De
.model De D (KF=3.204E-17,AF=1)
Ce 11 12 1GF
vse 12 0 dc 0V
he 1 vP vse 4.5k
Rhe 1 vP 1G
*innw = 0.566 fA/sqrt(Hz)
IDi 0 21 dc 1pA
Di 21 0 Di
.model Di D (KF=0)
Ci 21 22 1GF
vsi 22 0 dc 0V
fi 0 vN vsi 1u;***Zero in
*Noiseless op amp:
*a0 = 1V/uV, fb = 16 Hz, ft = 16 MHz
ea0 2 0 1 vN 1Meg
Req 2 3 1
Ceq 3 0 9.947mF
ebuf vO 0 3 0 1
.ends opa
*Main circuit:
Ii 2 1 ac 1nA
Vs 1 0 dc 0
R1 2 0 100G
C1 2 0 2nF
G2 2 4 2 4 27.4E-9; ***noiseless 36.5Meg
C2 2 3 0.5pF
R3 4 0 1k
R4 4 3 26.7k
X1 0 2 3 opa
.ac dec 10 0.1Hz 1GHz
.noise V(3) Vs 10
.probe
.end
Problem 7.35 (enoR)
.subckt opa vP vN vO
*enw = 4.5 nV/sqrt(Hz), fce = 100 Hz
IDe 0 11 dc 3.12uA
De 11 0 De
```

(7.38)

```
.model De D (KF=3.204E-17,AF=1)
Ce 11 12 1GF
vse 12 0 dc 0V
he 1 vP vse 4.5u;***zero en
Rhe 1 vP 1G
*innw = 0.566 fA/sqrt(Hz)
IDi 0 21 dc 1pA
Di 21 0 Di
.model Di D (KF=0)
Ci 21 22 1GF
vsi 22 0 dc 0V
fi 0 vN vsi 1u;***zero in
*Noiseless op amp:
*a0 = 1V/uV, fb = 16 Hz, ft = 16 MHz
ea0 2 0 1 vN 1Meg
Req 2 3 1
Ceq 3 0 9.947mF
ebuf vO 0 3 0 1
.ends opa
*Main circuit:
Ii 2 1 ac 1nA
Vs 1 0 dc 0
R1 2 0 100G
C1 2 0 2nF
R2 2 4 36.5Meg
C2 2 3 0.5pF
R3 4 0 1k
R4 4 3 26.7k
X1 0 2 3 opa
.ac dec 10 0.1Hz 1GHz
.noise V(3) Vs 10
.probe
.end
Problem 7.35 (enoi)
.subckt opa vP vN vO
*enw = 4.5 nV/sqrt(Hz), fce = 100 Hz
IDe 0 11 dc 3.12uA
De 11 0 De
.model De D (KF=3.204E-17,AF=1)
Ce 11 12 1GF
vse 12 0 dc 0V
he 1 vP vse 4.5u;***zero en
Rhe 1 vP 1G
*innw = 0.566 fA/sqrt(Hz)
IDi 0 21 dc 1pA
Di 21 0 Di
.model Di D (KF=0)
Ci 21 22 1GF
vsi 22 0 dc 0V
fi 0 vN vsi 1
*Noiseless op amp:
*a0 = 1V/uV, fb = 16 Hz, ft = 16 MHz
ea0 2 0 1 vN 1Meg
Req 2 3 1
Ceq 3 0 9.947mF
ebuf vO 0 3 0 1
.ends opa
*Main circuit:
```

7.39

```
Ii 2 1 ac 1nA
Vs 1 0 dc 0
R1 2 0 100G
C1 2 0 2nF
g2 2 4 2 4 27.4E-9;***noiseless 36.5Meg
C2 2 3 0.5pF
R3 4 0 1k
R4 4 3 26.7k
X1 0 2 3 opa
.ac dec 10 0.1Hz 1GHz
.noise V(3) Vs 10
.probe
.end
```



7.40

7.36




```
Problem 7.36
.subckt opa vP vN vO
*enw = 4.5 nV/sqrt(Hz), fce = 100 Hz
IDe 0 11 dc 3.12uA
De 11 0 De
.model De D (KF=3.204E-17,AF=1)
Ce 11 12 1GF
vse 12 0 dc 0V
he 1 vP vse 4.5k
Rhe 1 vP 1G
*innw = 0.566 fA/sqrt(Hz)
IDi 0 21 dc 1pA
Di 21 0 Di
.model Di D (KF=0)
Ci 21 22 1GF
vsi 22 0 dc 0V
fi 0 vN vsi 1
*Noiseless op amp:
*a0 = 1V/uV, fb = 16 Hz, ft = 16 MHz
ea0 2 0 1 vN 1Meg
Req 2 3 1
Ceq 3 0 9.947mF
ebuf vO 0 3 0 1
.ends opa
*Main circuit:
Ii 2 1 ac 1nA
Vs 1 0 dc 0
R1 2 0 100G
C1 2 0 2nF
R2 2 4 36.5Meg
C2 2 3 0.5pF
R3 4 0 1k
R4 4 3 26.7k
X1 0 2 5 opa
Cc 2 5 0.3pF
R5 5 3 5k
C3 3 0 100nF
.ac dec 10 0.1Hz 1GHz
.noise V(3) Vs 10
.probe
.end
```

(7.41)

Using a filter with $C_c = 0.3$ pF, $R_5 = 5\,k\Omega$, and $C_3 = 0.1\,\mu F$ reduces E_{no} from 1.5 mV to 530 μV at the price of a bandwidth reduction from 330 Hz to 236 Hz.



(7.42)

$\boxed{7.37}$ (a) Suming noise densities in RMS fashion gives $e_{no}^2 = (e_{n1}/N)^2 + (e_{n2}/N)^2 + \cdots (e_{nN}/N)^2 = N(e_n/N)^2 = e_n^2/N$, or $e_{no} = e_n/\sqrt{N}$.

(b) Imposing $e_{nR}^2 + e_n^2 \le (1.1e_n)^2$, or

[illegible]

we get $R \le$ [illegible] For instance, with $e_n = 10$ nV/$\sqrt{Hz}$ [illegible] 1.27 kΩ.